

純水と電解質水溶液の熱力学的性質を計算するプログラム (その1) —純水の計算プログラム—

Computer Programs for the Calculation of Thermodynamic Properties of Water and Aqueous Electrolyte Solutions. Part 1. A Computer Program for Pure Water

澁江 靖弘*
SHIBUE Yasuhiro

Haar 達 (Haar et al., 1984, NBS/NRC Steam Tables. Hemisphere Publishing) が発表した純水の熱力学的性質を計算する計算式と FORTRAN Code に基づいて BASIC/98 で記述した計算プログラムを作成した。内部エネルギーとエントロピーに関する基準状態である三重点での液相の値が0になるように Haar 達が与えた定数値を調節した。

キーワード：HGK 式, 水, 熱力学的性質

Keywords：HGK formulation, water, thermodynamic properties

1 はじめに

Haar 達 (Haar et al., 1984) が示した純水の状態方程式 (以下, HGK 式) とその FORTRAN プログラムを参考にして筆者は BASIC/98 を言語とするプログラムを作成した (澁江, 2005a, 2005b)。HGK 式が公表されて既に30年以上が経過しており, HGK 式よりも正確な状態方程式が既に公表されている。Wagner and Pruß (2002) が与えた式が正確に純水の熱力学的性質を与えるものとして知られている。さらに, Wagner and Pruß (2002) の式による計算結果を再現できる簡単な計算式が Wagner 達によって求められている (Wagner et al., 2000)。しかしながら, 電解質水溶液の熱力学的性質を求める時に必要な純水の性質は HGK 式で求められてきたことが多く, Wagner and Pruß (2002) や Wagner et al. (2000) が与えた式を用いて電解質水溶液の熱力学的性質を計算した報告は少ない。Mao and Duan (2008) の Wagner et al. (2000) を用いた報告や Zezin et al. (2014) の Wagner and Pruß (2002) を用いた報告などぐらいである。Pitzer (1995) によってまとめられた HGK 式を用いる報告に比べてはるかに少数である。そして, 筆者は電解質水溶液の熱力学的性質を HGK 式を用いて計算するプログラムを報告した (澁江, 2007a, 2007b)。また, 塩化ナトリウムや塩化カリウムを溶解している水溶液に関する状態方程式を利用する計算プログラムにも筆者は HGK 式を用いた (澁江, 2012a, 2012b)。電解質水溶液の性質の計算式を利用する上で HGK 式による計算プログラムは現在でも必要である。筆者は, 澁江 (2005a, 2005b) を公表した後も計算プログラムの改良を続けてきた (澁江, 2007a)。

これらの報告中で行った改良結果を踏まえて新たに作成した計算プログラムを報告する。

HGK 式では三重点 (273.16K, 0.00061173MPa) での液相をエネルギーの基準状態に取っている。つまり, この時の内部エネルギーとエントロピーの値が0になるようにしている。澁江 (2005a) が示した計算プログラムを筆者が現在使用しているコンピュータ (CPU が Intel Core i5-680) を用いて計算すると, 内部エネルギーが0.000028 J g⁻¹でエントロピーが-0.000004 J g⁻¹ K⁻¹となる。アルゴリズムによって三重点での内部エネルギーやエントロピーの値が厳密に0にならないことがあるので, Kestin et al. (1984) は定数の値を調節することを勧めている。そこで, 本研究では三重点での液相の内部エネルギーとエントロピーの値が0になるような調節を行う。

さて, 澁江 (2007b) は, HGK 式を用いて25℃で1atm における塩化ナトリウム水溶液中の塩化ナトリウムの部分モルエンタルピーを計算した時に標準状態での値が0にならなかったことを記した。この計算には HGK 式を用いて求めた純水のエンタルピーの値が必要になる。澁江 (2012b) は澁江 (2007b) が記した課題の解消方法について触れているが, 本研究で行う三重点での調節によってさらに塩化ナトリウムの部分モルエンタルピーの値が0に近づく。詳細については別に報告する予定である。

2 HGK 式

Haar 達が求めた純水の状態方程式 (ヘルムホルツエネルギー A を base 関数と residual 関数と ideal 関数から求

* 兵庫教育大学大学院教育実践高度化専攻理数系教科マネジメントコース 教授

められる値の和として与える式)を澁江(2005a)が示している。そして、澁江(2005b)中でヘルムホルツエネルギーとエントロピー、内部エネルギー、エンタルピー、ギブスエネルギー、定容熱容量、定圧熱容量の関係式を示している。これらの熱力学的性質を求める計算式を示すと長くなるので、ここでは省略する。

ヘルムホルツエネルギーを与える式から導くことができる圧力の計算式を表1中の式(1)として示す。式(1)中で用いている b , \bar{B} , y を与える式を表1中の式(2), 式(3), 式(4)として示す。

Haar 達が求めた646.3K 以下で適用可能な飽和蒸気圧(MPa)の計算式を表1中の式(5)あるいは式(6)として示す。式(5)は314K 以下で適用可能であり式(6)は314K より高温で適用可能な式である。なお、646.3K より高温条件での飽和蒸気圧の計算方法を澁江(2005b)が記しているため、ここでは省略する。

3 計算プログラム

水1 g 当たりの内部エネルギーを気体定数(水1 g 当たりに換算した値で、プログラム中では変数 GASCON)と絶対温度(変数名 T)の積で割って無次元化した値を変数 UD としてプログラムでは使用している。UD の値は base 関数から求められる無次元化した内部エネルギー(変数名 UB), residual 関数から求められる無次元化した内部エネルギー(変数名 UR), ideal 関数から求められる無次元化した内部エネルギー(変数名 UI)と三重点で0になるようにするための定数(変数名 UREF)と絶対温度から求めている。計算式は次の通りである。

$$UD = UB + UR + UI - UREF/T$$

273.16K での気液平衡条件を澁江(2005a)中の計算プログラムで求めると三重点での液相に関する UD の値が 2.2543×10^{-7} であった。すると273.16を 2.2543×10^{-7} にかけて求められる 6.1578×10^{-5} を UREF に加えるならば UD の値は0に極めて近くなる。Haar 達は UREF の値を -4.328455039×10^3 と与えているので 6.2×10^{-5} を加えて調節後の UREF の値を -4.328454977×10^3 とした。

水1 g 当たりのエントロピーを GASCON で割って無次元化した値を変数 SD としてプログラムでは使用している。SD の値は base 関数から求められる無次元化したエントロピー(変数名 SB), residual 関数から求められる無次元化したエントロピー(変数名 SR), ideal 関数から求められる無次元化したエントロピー(変数名 SI)と三重点で0になるようにするための定数(変数名 SREF)から求めている。計算式は次の通りである。

$$SD = SB + SR + SI - SREF$$

273.16K での気液平衡条件を澁江(2005a)中の計算プログラムで求めると三重点での液相に関する SD の値は -8.1833×10^{-6} であった。SREF にこの値を加えるならば SD の値は0に極めて近くなる。Haar 達は SREF の値を7.6180802と与えているので -8.2×10^{-6} を加えて調節後の SREF の値を7.6180720とした。UREF と SREF の値を変えたことが三重点での液相の密度や飽和蒸気圧の計算結果に影響を与えていないことを確認している。

これら以外の改良点や修正点は計算結果に影響を及ぼさないものだけであるので、簡単に箇条書きで記す。

澁江(2005a)や澁江(2005b)中の計算プログラムで不要な変数や計算命令があるので削除した。また、いくつかの変数名を改めるとともに一部のサブルーチン名を短縮した。

単位換算のための計算と計算結果の印字を、それぞれ、サブルーチン *UNIT_CONVERT とサブルーチン *WATER_PROPERTY としてまとめた。

温度と圧力を入力して密度を求める計算での収束条件(密度の推定値から計算できる圧力と入力圧力との食い違いに関する収束条件)を厳しくした。同様のことを飽和蒸気圧の計算で気液平衡が成立しているのかどうかの判定条件(気相と液相のギブスエネルギーが等しくなっている条件)を厳しくした。

澁江(2005b, pp. 144-145)が記したように Haar 達が与えた臨界温度647.126K まで気液平衡計算を行えるように改めた。

単位として K (プログラムでは「kelvin」)か°C (プログラムでは「deg C」), kg m^{-3} (プログラムでは「kg/m3」)か g cm^{-3} (プログラムでは「g/cm3」), 「MPa」か「bar」, 「kJ/kg」か「J/g」を入力して選択するようにした。そして、出力値については、温度以外は指数表記で表すようにした。計算値の有効桁数が5桁である場合が多いが、四捨五入による丸めの影響を知るために6桁目も出力させるように改めた。

サブルーチン *PST 中で使用する定数値, サブルーチン *DFIND 中で使用するファクター1.02と0.98をサブルーチン *BLOCKDATA 内で読み込むことにした。同様にサブルーチン *BASEDT 中で用いている0.101325の値をサブルーチン *BLOCKDATA で読み込むことにした。

以上の改良を加えて新たに本研究で作成した計算プログラムを表2に示す。そして、気液平衡条件での計算プログラムを表3に示す。表2と表3のいずれについても50刻みで行番号を取っている。表3中では表2と同じサブルーチンを使用する場合には各サブルーチンの最初の行番号とサブルーチン名だけを示している。表2に示したサブルーチンの内容を GOTO 文で指定する行番号を改めて、演算内容を写すことでプログラム全体を作成す

表1 圧力 p を密度 d_w と絶対温度 T で表す式および飽和蒸気圧 p_{sat} の計算式*(Haar et al., 1984)

$$p = d_w RT \left[\frac{1 + \alpha y + \beta y^2}{(1-y)^3} + 4y \left(\frac{\bar{B}}{b} - \gamma \right) \right] + d_w^2 \sum_{i=1}^{36} g_i \left(\frac{647.073}{T} \right)^{l_i} e^{-d_w} (1 - e^{-d_w})^{k_i - 1} + d_w^2 \sum_{i=37}^{40} \frac{g_i}{\rho_i} \left(\frac{d_w - \rho_i}{\rho_i} \right)^{l_i} \left[l_i \left(\frac{d_w - \rho_i}{\rho_i} \right)^{-1} - \alpha_i k_i \left(\frac{d_w - \rho_i}{\rho_i} \right)^{k_i - 1} \right] \exp \left[-\alpha_i \left(\frac{d_w - \rho_i}{\rho_i} \right)^{k_i} - \beta_i \left(\frac{T - T_i}{T_i} \right)^2 \right] \quad (1)$$

$$b = b_0 + b_1 \ln \left(\frac{T}{647.073} \right) + b_3 \left(\frac{647.073}{T} \right)^3 + b_5 \left(\frac{647.073}{T} \right)^5 \quad (2)$$

$$\bar{B} = \sum_{j=0}^5 B_j \left(\frac{647.073}{T} \right)^j \quad (3)$$

$$y = \frac{b d_w}{4} \quad (4)$$

$$p_{\text{sat}} = 0.1 \cdot \exp \left(6.3573118 - 8858.8430 / T + 607.56335 T^{-0.6} \right) \quad (5) \quad (T \leq 314)$$

$$\ln(p_{\text{sat}} / 22.093) = \frac{647.25}{T} \left[A_1 \left(1 - \frac{T}{647.25} \right) + A_2 \left(1 - \frac{T}{647.25} \right)^{3/2} + A_3 \left(1 - \frac{T}{647.25} \right)^2 + A_4 \left(1 - \frac{T}{647.25} \right)^{5/2} \right] + \frac{647.25}{T} \left[A_5 \left(1 - \frac{T}{647.25} \right)^3 + A_6 \left(1 - \frac{T}{647.25} \right)^{7/2} \right] + \frac{647.25}{T} \left[A_7 \left(1 - \frac{T}{647.25} \right)^4 + A_8 \left(1 - \frac{T}{647.25} \right)^{9/2} \right] \quad (6) \quad (314 < T \leq 646.3)$$

* 計算式中の記号の意味を本文中あるいは澁江(2005a, 2005b)中で示している。計算式中の係数値や定数値はプログラム中の変数名と関連付けて後で示す。

ることができる。例えば、表3中で line 20400として *PCORR と記している。これは表2中の line 21900を line 20400として使用し、表2中の line 21950から line 22300を表3中で line 20450から line 20800として補うことを意味している。このようにすると、表2の line 22200中に含まれている GOTO 文は表3では line 20700に含まれている GOTO 20800の命令文に改めることになる。同様に表2中の line 22250中の GOTO 文は表3では line 20750に含まれている GOTO 20550の命令文に改めることになる。その他のサブルーチンについても同様の処理を施すことによって表3で示したプログラム中の省略した箇所を復元することができる。なお、表2中に#が付いていない数値があるが、計算では倍精度型数値として取り扱われている。

プログラムを走らせた後の入力方法と出力結果は澁江(2005a, 2005b)中で示したものと単位の選択方法や出力値の表記を改めた点を除けば同じである。繰り返しになる箇所が大部分であるので省略する。

表4はHGK式中で用いられている定数値と係数値を表す変数名を示す。表5と表6は表2や表3中で使用しているサブルーチンでの計算内容などを示す。表5はサブルーチン *DFIND, *IDEALT, *BBT, *BASEDT, *QQT, *THERMDTについて記し、表6はサブルーチン *PCORR, *PST, *CORR, *UNIT, *UNIT_CONVERT, *TTTT, *WATER_PROPERTY, *BLOCKDATAについて記している。

表2 HGK式による純水の性質を計算するプログラム

```

10000 REM HGK
10050 DEFDBL A-H, M-Z
10100 DIM HGKG(40),II(40),JJ(40),BP(7),BQ(7),AST(5)
10150 DIM ATZ(4),ADZ(4),AAT(4),AAD(4)
10200 DIM BR(6),A(8),HGKC(18)
10250 DIM QR(11),QT(10),QZR(9),QZT(9)
10300 DIM FFD(2),FFP(2),FFH(2),NNT$(2),NND$(2),NNP$(2),NNH$(2)
10350 GOSUB *BLOCKDATA
10400 GOSUB *UNIT
10450 PRINT
10500 PRINT "Enter option and temperature.  Option=1, input density.  Option=2, input pressure"
10550 INPUT "Option No. ";IOPT
10600 IF IOPT<>1 AND IOPT<>2 THEN PRINT:GOTO 10550
10650 INPUT "Density or Pressure";X
10700 INPUT "Temperature";TT
10750 GOSUB *TTTT
10800 RT=GASCON*T
10850 GOSUB *BBT
10900 ON IOPT GOTO 10950, 11250
10950 DD=X
11000 D=DD*FD
11050 GOSUB *THERMDT
11100 PRES=FP*(RT*D*ZBASE+Q)
11150 DOUT=DD*FD
11200 GOTO 11900
11250 PRES=X
11300 PINPUT=PRES/FP
11350 DGSS=PINPUT/(T*.4#)
11400 IF T>=647.126# THEN GOTO 11650
11450 DLL=0#:DVV=0#:DLIQ=0#:DVAP=0#
11500 GOSUB *PCORR
11550 IF ABS((PINPUT-PPP)/PPP)=<5D-005 THEN PPP=PINPUT:GOTO 12400
11600 IF PINPUT>PPP THEN DGSS=DL
11650 D=DGSS:PPP=PINPUT
11700 GOSUB *DFIND
11750 D=DOUT
11800 GOSUB *THERMDT
11850 DD=DOUT/FD
11900 GOSUB *UNIT_CONVERT
11950 LPRINT "Units"
12000 LPRINT A1$;SPC(3);NT$
12050 LPRINT A2$;SPC(3);ND$
12100 LPRINT A3$;SPC(3);NP$
12150 LPRINT A4$;SPC(3);NH$
12200 LPRINT
12250 GOSUB *WATER_PROPERTY
12300 LPRINT:LPRINT
12350 GOTO 13600
12400 D=DL
12450 IF T>646.3# THEN DOUT=D:GOTO 12600
12500 GOSUB *DFIND
12550 D=DOUT
12600 GOSUB *THERMDT
12650 GOSUB *UNIT_CONVERT
12700 PRES=PPP*FP
12750 LPRINT "Units"
12800 LPRINT A1$;SPC(3);NT$
12850 LPRINT A2$;SPC(3);ND$
12900 LPRINT A3$;SPC(3);NP$
12950 LPRINT A4$;SPC(3);NH$
13000 LPRINT
13050 LPRINT "Liquid phase"
13100 GOSUB *WATER_PROPERTY
13150 D=DV
13200 IF T>646.3# THEN DOUT=D:GOTO 13350
13250 GOSUB *DFIND
13300 D=DOUT
13350 GOSUB *THERMDT
13400 GOSUB *UNIT_CONVERT
13450 LPRINT "Vapor phase"
13500 GOSUB *WATER_PROPERTY
13550 LPRINT:LPRINT
13600 INPUT "Will you continue the calculation?  Input Y(or y) or N(or n)";CAL$
13650 IF CAL$="Y" OR CAL$="y" THEN GOTO 10450
13700 END
13750 *DFIND
13800 DD=D
13850 LL=0
13900 LL=LL+1
13950 IF DD=<1D-008 THEN DD=1D-008
14000 IF DD>1.9# THEN DD=1.9#

```

純水と電解質水溶液の熱力学的性質を計算するプログラム (その1)

```

14100 GOSUB *QQTD
14150 GOSUB *BASEDT
14200 PP=RT*DD*ZBASE+Q
14250 DPDD=RT*(ZBASE+Y*DZB)+Q5
14300 IF DPDD>0 THEN GOTO 14500
14350 IF D>=.2967# THEN DD=DD*DINC
14400 IF D<.2967# THEN DD=DD*DDEC
14450 IF LL=<10 GOTO 13900
14500 DPDX=DPDD*1.1#
14550 IF DPDX<.1# THEN DPDX=.1#
14600 PPERR=ABS(1#-PP/PPP)
14650 IF PPERR<1D-009 THEN GOTO 15050
14700 IF D>.3# AND PPERR<1D-008 THEN GOTO 15050
14750 IF D>.7# AND PPERR<1D-007 THEN GOTO 15050
14800 XP=(PPP-PP)/DPDX
14850 IF ABS(XP)>.1# THEN XP=XP*.1#/ABS(XP)
14900 DD=DD+XP
14950 IF DD=<0 THEN DD=1D-008
15000 IF LL=<30 THEN GOTO 13900
15050 DOUT=DD
15100 RETURN
15150 *IDEALT
15200 TIDEAL=T/100
15250 TIL=LOG(TIDEAL)
15300 GI=(-1#)*(HGKC(1)/TIDEAL+HGKC(2))*TIL
15350 HI=(HGKC(2)+HGKC(1)*(1#-TIL)/TIDEAL)
15400 CPI=HGKC(2)-HGKC(1)/TIDEAL
15450 FOR I=3 TO 18
15500 TIDE=TIDEAL^CDBL(I-6)
15550 GI=GI-HGKC(I)*TIDE
15600 HI=HI+HGKC(I)*CDBL(I-6)*TIDE
15650 CPI=CPI+HGKC(I)*CDBL(I-6)*CDBL(I-5)*TIDE
15700 NEXT I
15750 AI=GI-1#
15800 UI=HI-1#
15850 CVIX=CPI-1#
15900 SI=UI-AI
15950 RETURN
16000 *BBT
16050 BR(1)=1#
16100 FOR I=2 TO 6
16150 BR(I)=BR(I-1)*TZ/T
16200 NEXT I
16250 B1=BP(1)+BP(2)*LOG(1#/BR(2))
16300 B2=BQ(1)
16350 B1T=BP(2)*BR(2)/TZ
16400 B2T=0
16450 B1TT=0
16500 B2TT=0
16550 FOR I=3 TO 7
16600 B1=B1+BP(I)*BR(I-1)
16650 B2=B2+BQ(I)*BR(I-1)
16700 B1T=B1T-CDBL(I-2)*BP(I)*BR(I-1)/T
16750 B2T=B2T-CDBL(I-2)*BQ(I)*BR(I-1)/T
16800 B1TT=B1TT+BP(I)*CDBL(I-2)*CDBL(I-2)*BR(I-1)/(T*T)
16850 B2TT=B2TT+BQ(I)*CDBL(I-2)*CDBL(I-2)*BR(I-1)/(T*T)
16900 NEXT I
16950 B1TT=B1TT-B1T/T
17000 B2TT=B2TT-B2T/T
17050 RETURN
17100 *BASEDT
17150 Y=.25#*B1*D
17200 XX=1#-Y
17250 Z0=(1#+ALPHAHGK*Y+BETAHGK*Y*Y)/(XX*XX*XX)
17300 ZBASE=Z0+4#*Y*(B2/B1-GAMMAHGK)
17350 DZ0=(ALPHAHGK+2#*BETAHGK*Y)/(XX*XX*XX)+3#*Z0/XX
17400 DZB=DZ0+4#*(B2/B1-GAMMAHGK)
17450 AB=(-1#)*LOG(XX)-(BETAHGK-1#)/XX+28.16666667#/(XX*XX)
17500 AB=AB+4#*Y*(B2/B1-GAMMAHGK)+15.166666667#+LOG(D*RT/P0)
17550 BB2TT=T*T*B2TT
17600 UB=(-1#)*T*B1T*(ZBASE-1#-D*B2)/B1-D*T*B2T
17650 CVB=2#*UB+(Z0-1#)*((T*B1T/B1)*(T*B1T/B1)-T*T*B1TT/B1)
17700 CVB=CVB-D*(BB2TT-GAMMAHGK*B1TT*T*T)-(T*B1T/B1)*(T*B1T/B1)*Y*DZ0
17750 DPDTB=ZBASE/T+D*(DZB*B1T/4#+B2T-B2*B1T/B1)
17800 SB=UB-AB
17850 RETURN
17900 *QQTD
17950 QR(1)=0
18000 Q5=0
18050 Q=0
18100 AR=0
18150 DADT=0
18200 DPDTR=0
18250 CVR=0#

```

```

18350 Q10=D*D*E
18400 Q20=1#-E
18450 QR(2)=Q10
18500 QV=TZ/T
18550 QT(1)=T/TZ
18600 FOR I=2 TO 10
18650 QR(I+1)=QR(I)*Q20
18700 QT(I)=QT(I-1)*QV
18750 NEXT I
18800 FOR I=1 TO 36
18850 K=II(I)+1
18900 L=JJ(I)
18950 QK=CDBL(K):QL=CDBL(L)
19000 QZR(K-1)=QR(K+1):QZT(L)=QT(L+1):QZR(K)=QR(K+2):QZT(L+1)=QT(L+2)
19050 QP=HGKG(I)*QZR(K-1)*QZT(L)
19100 Q=Q+QP
19150 Q5=Q5+(2#/D-(1#-E*(QK-1#)/Q20))*QP
19200 AR=AR+HGKG(I)*QZR(K)*QZT(L)/(Q10*QK*RT)
19250 DFDT=(Q20^QK)*(1#-QL)*QZT(L+1)/(TZ*QK)
19300 D2F=QL*DFDT
19350 DPT=DFDT*Q10*QK/Q20
19400 DADT=DADT+HGKG(I)*DFDT
19450 DPDTR=DPDTR+HGKG(I)*DPT
19500 CVR=CVR+HGKG(I)*D2F/GASCON
19550 NEXT I
19600 Q2A=0
19650 FOR J=37 TO 40
19700 K=II(J)
19750 KM=JJ(J)
19800 QK=CDBL(K):QKM=CDBL(KM)
19850 DDZ=ADZ(J-36)
19900 DEL=D/DDZ-1#
19950 IF ABS(DEL)<1D-010 THEN DEL=1D-010
20000 EX1=(-1#)*AAD(J-36)*(DEL^QK)
20050 DEX=EXP(EX1)*(DEL^QKM)
20100 ATT=AAT(J-36)
20150 TX=ATZ(J-36)
20200 TAU=T/TX-1#
20250 EX2=(-1#)*ATT*TAU*TAU
20300 TEX=EXP(EX2)
20350 Q30=DEX*TEX
20400 QM=QKM/DEL-QK*AAD(J-36)*(DEL^(QK-1#))
20450 FCT=QM*D*D*Q30/DDZ
20500 Q5T=FCT*(2#/D+QM/DDZ)
20550 Q5T=Q5T-(D/DDZ)*(D/DDZ)*Q30*(QKM/(DEL*DEL)+QK*(QK-1#)*AAD(J-36)*(DEL^(QK-2#)))
20600 Q5=Q5+Q5T*HGKG(J)
20650 Q=Q+HGKG(J)*FCT
20700 DADT=DADT-2#*HGKG(J)*ATT*TAU*Q30/TX
20750 DPDTR=DPDTR-2#*HGKG(J)*ATT*TAU*FCT/TX
20800 Q2A=Q2A+T*HGKG(J)*(4#*ATT*EX2+2#*ATT)*Q30/(TX*TX)
20850 AR=AR+Q30*HGKG(J)/RT
20900 NEXT J
20950 SR=(-1#)*DADT/GASCON
21000 UR=AR+SR
21050 CVR=CVR+Q2A/GASCON
21100 RETURN
21150 *THERMDT
21200 GOSUB *IDEALT
21250 GOSUB *BASEDT
21300 GOSUB *QQTD
21350 Z=ZBASE+Q/(RT*D)
21400 DPDD=RT*(ZBASE+Y*DZB)+Q5
21450 AD=AB+AR+AI-UREF/T+SREF
21500 GD=AD+Z
21550 UD=UB+UR+UI-UREF/T
21600 DPDT=RT*D*DPDTB+DPDTR
21650 CVDX=CVB+CVR+CVIX
21700 CPD=CVDX+T*DPDT*DPDT/(D*D*DPDD*GASCON)
21750 HD=UD+Z
21800 SD=SB+SR+SI-SREF
21850 RETURN
21900 *PCORR
21950 GOSUB *PST
22000 PPP=PS
22050 GOSUB *CORR
22100 DP=DELG*RT/(1#/DV-1#/DL)
22150 PPP=PPP+DP
22200 IF ABS(DELG)<.00001# THEN GOTO 22300
22250 DLL=DL:DVV=DV:GOTO 22050
22300 RETURN
22350 *PST
22400 IF T>314# THEN GOTO 22600
22450 PL=AST(1)-AST(2)/T+AST(3)*(T^(-.6#))
22500 PS=.1#*EXP(PL)

```

純水と電解質水溶液の熱力学的性質を計算するプログラム (その1)

```

22600 W=ABS(1#-T/AST(4))
22650 BPST=0#
22700 FOR I=1 TO 8
22750 BPST=BPST+A(I)*(W^((CDBL(I)+1#)/2#))
22800 NEXT I
22850 QPST=BPST*(AST(4)/T)
22900 PS=AST(5)*EXP(QPST)
22950 RETURN
23000 *CORR
23050 IF T>646.3# THEN GOTO 23950
23100 DLIQ=DLL
23150 IF DLL=<0 THEN DLIQ=1.11#-.0004*T
23200 DLL=DLIQ:D=DLIQ
23250 GOSUB *DFIND
23300 D=DOUT:DL=DOUT
23350 GOSUB *THERMDT
23400 GL=GD
23450 DVAP=DVV
23500 IF DVV=<0 THEN DVAP=PPP/RT
23550 DVV=DVAP:D=DVAP
23600 GOSUB *DFIND
23650 IF DOUT<5D-007 THEN DOUT=5D-007
23700 D=DOUT:DV=DOUT
23750 GOSUB *THERMDT
23800 GV=GD
23850 DELG=GL-GV
23900 RETURN
23950 PPP=0
24000 DELG=0
24050 DC=.657128#*((1#-T/647.126#)^.325#)
24100 DL=.322#+DC
24150 DV=.322#-DC
24200 D=DV
24250 GOSUB *BASEDT
24300 GOSUB *QQTD
24350 PPP=RT*D*DV*ZBASE+Q
24400 RETURN
24450 *UNIT
24500 PRINT"*****"
24550 PRINT"* Enter units      *"
24600 PRINT"*****"
24650 PRINT A1$
24700 PRINT"Choose from 1=kelvin, 2=deg C"
24750 INPUT IT
24800 IF IT<1 OR IT>2 THEN GOTO 24700
24850 NT$=NNT$(IT)
24900 PRINT A2$
24950 PRINT"Choose from 1=kg/m3, 2=g/cm3"
25000 INPUT ID
25050 IF ID>2 OR ID<1 THEN GOTO 24950
25100 ND$=NND$(ID)
25150 FD=FFD(ID)
25200 PRINT A3$
25250 PRINT"Choose from 1=MPa, 2=bar"
25300 INPUT IP
25350 IF IP>2 OR IP<1 THEN GOTO 25250
25400 NP$=NNP$(IP)
25450 FP=FFP(IP)
25500 PRINT A4$
25550 PRINT"Choose from 1=kJ/kg, 2=J/g"
25600 INPUT IH
25650 IF IH>2 OR IH<1 THEN GOTO 25550
25700 NH$=NNH$(IH)
25750 FH=FFH(IH)
25800 RETURN
25850 *UNIT_CONVERT
25900 DD=DOUT/FD
25950 U=UD*RT*FH
26000 H=HD*RT*FH
26050 S=SD*GASCON*FH
26100 CP=CPD*GASCON*FH
26150 CV=CVDX*GASCON*FH
26200 DPDD=DPDD*FD*FP
26250 DPDT=DPDT*FP
26300 G=GD*RT*FH
26350 A=AD*RT*FH
26400 RETURN
26450 *TTTT
26500 ON IT GOTO 26550, 26650
26550 T=TT
26600 GOTO 26700
26650 T=TT+273.15#
26700 RETURN
26750 *WATER_PROPERTY

```

```

26850 LPRINT USING"DP/DT= +#.#####^" DP/DD= +#.#####^";DPDT,DPDD
26900 LPRINT USING"CP= +#.#####^" CV= +#.#####^";CP,CV
26950 LPRINT USING"S= +#.#####^" H= +#.#####^ U= +#.#####^";S,H,U
27000 LPRINT USING"G= +#.#####^" A= +#.#####^";G,A
27050 RETURN
27100 *BLOCKDATA
27150 FOR I=1 TO 4:READ ATZ(I):NEXT I
27200 DATA 640#,640#,641.6#,270#
27250 FOR I=1 TO 4:READ ADZ(I):NEXT I
27300 DATA 0.319#,0.319#,0.319#,1.55#
27350 FOR I=1 TO 4:READ AAT(I):NEXT I
27400 DATA 2.0D+004,2.0D+004,4.0D+004,25.0#
27450 FOR I=1 TO 4:READ AAD(I):NEXT I
27500 DATA 34.0#,40.0#,30.0#,1.05D+003
27550 GASCON=-.461522#:TZ=6.47073D+002
27600 UREF=-4328.454977#:SREF=7.6180720D+000
27650 ALPHAHGK=11#:BETAHGK=44.333333333333#:GAMMAHGK=3.5#
27700 FOR I=1 TO 7:READ BP(I):NEXT I
27750 DATA 0.7478629#,-0.3540782#,0.0#,0.0#,0.007159876#,0.0#,-0.003528426#
27800 FOR I=1 TO 7:READ BQ(I):NEXT I
27850 DATA 1.1278334#,0.0#,-0.5944001#,-5.010996#,0.0#,0.63684256#,0.0#
27900 FOR I=1 TO 40:READ HGKG(I):NEXT I
27950 DATA -5.3062968529023D+002,2.2744901424408D+003,7.8779333020687D+002
28000 DATA -6.9830527374994D+001,1.7863832875422D+004,-3.9514731563338D+004
28050 DATA 3.3803884280753D+004,-1.3855050202703D+004,-2.5637436613260D+005
28100 DATA 4.8212575981415D+005,-3.4183016969660D+005,1.2223156417448D+005
28150 DATA 1.1797433655832D+006,-2.1734810110373D+006,1.0829952168620D+006
28200 DATA -2.5441998064049D+005,-3.1377774947767D+006,5.2911910757704D+006
28250 DATA -1.3802577177877D+006,-2.5109914369001D+005,4.6561826115608D+006
28300 DATA -7.2752773275387D+006,4.1774246148294D+005,1.4016358244614D+006
28350 DATA -3.1555231392127D+006,4.7929666384584D+006,4.0912664781209D+005
28400 DATA -1.3626369388386D+006,6.9625220862664D+005,-1.0834900096447D+006
28450 DATA -2.2722827401688D+005,3.836548600660D+005,6.8833257944332D+003
28500 DATA 2.1757245522644D+004,-2.6627944829770D+003,-7.0730418082074D+004
28550 DATA -0.225#,-1.68#,0.055#,-93.0#
28600 FOR I=1 TO 40:READ II(I):NEXT I
28650 DATA 0,0,0,1,1,1,1,2,2,2,2,3,3,3,3,4,4,4,4,5,5,5,5,6,6,6,6,8,8,8,8,2,2,0,4,2,2,2,4
28700 FOR I=1 TO 40:READ JJ(I):NEXT I
28750 DATA 2,3,5,7,2,3,5,7,2,3,5,7,2,3,5,7,2,3,5,7,2,3,5,7,2,3,5,7,1,4,4,4,0,2,0,0
28800 FOR I=1 TO 8:READ A(I):NEXT I
28850 DATA -7.8889166#,2.5514255#,-6.716169#,33.239495#
28900 DATA -105.38479#,174.35319#,-148.39348#,48.631602#
28950 FOR I=1 TO 18:READ HGKC(I):NEXT I
29000 DATA 1.9730271018D+001,2.09662681977D+001,-4.83429455355D-001,6.05743189245D+000
29050 DATA 2.256023885D+001,-9.87532442D+000,-4.3135538513D+000,4.58155781D-001
29100 DATA -4.7754901883D-002,4.1238460633D-003,-2.7929052852D-004
29150 DATA 1.4481695261D-005,-5.6473658748D-007,1.6200446D-008,-3.303822796D-010
29200 DATA 4.51916067368D-012,-3.70734122708D-014,1.37546068238D-016
29250 FOR I=1 TO 5:READ AST(I):NEXT I
29300 DATA 6.3573118#,8858.843#,6.0756335D+002,647.25#,2.2093D+001
29350 DINC=1.02#:DDEC=0.98#
29400 P0=1.01325D-001
29450 FOR I=1 TO 2:READ FFD(I):NEXT I
29500 DATA 1.0D-003,1.0#
29550 FOR I=1 TO 2:READ FFP(I):NEXT I
29600 DATA 1.0#,10.0#
29650 FOR I=1 TO 2:READ FFH(I):NEXT I
29700 DATA 1.0#,1.0#
29750 FOR I=1 TO 2:READ NNT$(I):NEXT I
29800 DATA "K","deg C"
29850 FOR I=1 TO 2:READ NND$(I):NEXT I
29900 DATA "kg/m3","g/cm3"
29950 FOR I=1 TO 2:READ NNP$(I):NEXT I
30000 DATA "MPa","bar"
30050 FOR I=1 TO 2:READ NNH$(I):NEXT I
30100 DATA "kJ/kg","J/g"
30150 A1$="Temperature":A2$="Density":A3$="Pressure":A4$="Energy"
30200 RETURN

```

表3 気液二相平衡条件下でのHGK式による純水の性質を計算するプログラム#

```

10000 REM HGK Vapor-liquid equilibrium.
10050 DEFDBL A-H, M-Z
10100 DIM HGKG(40),II(40),JJ(40),BP(7),BQ(7),AST(5)
10150 DIM ATZ(4),ADZ(4),AAT(4),AAD(4)
10200 DIM BR(6),A(8),HGKC(18)
10250 DIM QR(11),QT(10),QZR(9),QZT(9)
10300 DIM FFD(2),FFP(2),FFH(2),NNT$(2),NND$(2),NNP$(2),NNH$(2)
10350 GOSUB *BLOCKDATA
10400 GOSUB *UNIT
10450 PRINT
10500 PRINT "Enter temperature"
10550 INPUT"Temperature";TT
10600 GOSUB *TTTT
10650 IF T>=647.126# THEN PRINT "Input temperature>=critical temperature.":GOTO 10550
10700 RT=GASCON*T
10750 GOSUB *BBT
10800 DLL=0#:DVV=0#:DLIQ=0#:DVAP=0#
10850 GOSUB *PCORR
10900 D=DL
10950 IF T>646.3# THEN DOUT=D:GOTO 11100
11000 GOSUB *DFIND
11050 D=DOUT
11100 GOSUB *THERMDT
11150 GOSUB *UNIT_CONVERT
11200 PRES=PPP*FP
11250 LPRINT"Units"
11300 LPRINT A1$;SPC(3);NT$
11350 LPRINT A2$;SPC(3);ND$
11400 LPRINT A3$;SPC(3);NP$
11450 LPRINT A4$;SPC(3);NH$
11500 LPRINT
11550 LPRINT"Liquid phase"
11600 GOSUB *WATER_PROPERTY
11650 D=DV
11700 IF T>646.3# THEN DOUT=D:GOTO 11850
11750 GOSUB *DFIND
11800 D=DOUT
11850 GOSUB *THERMDT
11900 GOSUB *UNIT_CONVERT
11950 LPRINT"Vapor phase"
12000 GOSUB *WATER_PROPERTY
12050 LPRINT:LPRINT:LPRINT
12100 INPUT"Will you continue the calculation? Input Y(or y) or N(or n)":CAL$
12150 IF CAL$="Y" OR CAL$="y" THEN GOTO 10450
12200 END
12250 *DFIND
13650 *IDEALT
14500 *BBT
15600 *BASEDT
16400 *QQTD
19650 *THERMDT
20400 *PCORR
20850 *PST
21500 *CORR
22950 *UNIT
24350 *UNIT_CONVERT
24950 *TTTT
25250 *WATER_PROPERTY
25600 *BLOCKDATA

```

サブルーチンである*DFIND, *IDEALT, *BBT, *BASEDT, *QQTD, *THERMDT, *PCORR, *PST, *CORR, *UNIT, *UNIT_CONVERT, *TTTT, *WATER_PROPERTY, *BLOCKDATAの内容は表2中で示したものと同一である。ただしサブルーチン中のGOTO文で指定する行番号を改めておく必要がある。

表 4 HGK 式中で使用されている定数値や係数値
のプログラム中での変数名

変数の値あるいは意味	変数名
314K 以下での飽和蒸気圧の近似式の係数 [式(5)]	AST(1), AST(2), AST(3)
a_i [式(1)]	AAD(J-36)
β_i [式(1)]	AAT(J-36)
ρ_i [式(1)]	ADZ(J-36)
α [式(1)]	ALPHAHGK
314K より高温での飽和蒸気圧の近似式の係数 A_i と定数 22.093 および係数 647.25 [式(6)]	A(I), AST(4), AST(5)
T_i [式(1)]	ATZ(J-36)
β [式(1)]	BETAHGK
b_0 [式(2)]	BP(1)
b_1 [式(2)]	BP(2)
0 [式(2)]	BP(3)
0 [式(2)]	BP(4)
b_3 [式(2)]	BP(5)
0 [式(2)]	BP(6)
b_5 [式(2)]	BP(7)
B_0 [式(3)]	BQ(1)
0 [式(3)]	BQ(2)
B_1 [式(3)]	BQ(3)
B_2 [式(3)]	BQ(4)
B_3 [式(3)]	BQ(5)
B_4 [式(3)]	BQ(6)
B_5 [式(3)]	BQ(7)
0.98 (サブルーチン*DFIND で密度の推定値を改良するための値)	DDEC
1.02 (サブルーチン*DFIND で密度の推定値を改良するための値)	DINC

(右段に続く)

変数の値あるいは意味	変数名
密度の単位を換算するための係数。計算過程では g cm^{-3} を単位にしている。	FFD(I)
エネルギーの単位を換算するための係数。計算過程では J g^{-1} を単位にしている。	FFH(I)
圧力の単位を換算するための係数。計算過程では MPa を単位にしている。	FFP(I)
γ [式(1)]	GAMMAHGK
気体定数の値を水のモル質量で割って求めた値 R [式(1)]	GASCON
理想気体のヘルムホルツエネルギーの計算式の係数。澁江 (2005a) 中の表 1-3 中に計算式を示している。	HGKC(I)
g_i [式(1)]	HGKG(I)
$k_i - 1$ ($i = 1, \dots, 36$) あるいは k_i ($i = 37, \dots, 40$) [式(1)]	II(I)
$l_i + 1$ ($i = 1, \dots, 36$) あるいは l_i ($i = 37, \dots, 40$) [式(1)]	JJ(I)
0.101325 MPa	P0
三重点での液相のエントロピーを 0 にするための定数値	SREF
647.073*	TZ
三重点での液相の内部エネルギーを 0 にするための定数値	UREF

*式(1), 式(2), 式(3)中では変数 TZ の値(647.073) を使って示している。

表 5 サブルーチン*DFIND, *IDEALT, *BBT, *BASEDT, *QQTD, *THERMDTでの計算内容

サブルーチン*DFIND

温度と圧力から密度 (変数名 DOUT) を計算する。密度の推定値 (変数名 D) から求められる圧力の計算値 (変数名 PP) と圧力の入力値 (変数名 PPP) から収束判定式の値 $|1 - PP/PPP|$ を計算する。この値が収束条件を満たしている時に密度が求められたとする。収束条件は表 2 中の line 14650 から line 14750 で示したものである。DPDD (圧力の密度 d_w に関する偏導関数の値) の値が 0 以下になる時, line 14350 と line 14400 において d_w が 0.2967 g cm^{-3} 以上の時には d_w に DINC をかけ, 0.2967 g cm^{-3} より小さい時には d_w に DDEC をかけて密度の推定値を改めている。

サブルーチン*IDEALT

温度から理想気体状態におけるヘルムホルツエネルギー (変数名 AI), ギブスエネルギー (変数名 GI), エントロピー (変数名 SI), 内部エネルギー (変数名 UI), エンタルピー (変数名 HI), 定容熱容量 (変数名 CVIX), 定圧熱容量 (変数名 CPI) の値を計算する。いずれの値も気体定数あるいは気体定数と絶対温度の積で割って無次元化した値として求めている。定容熱容量に関する変数を CVI にした方が他の変数と調和的になるが, CVI は BASIC/98 の予約語になっているので使用していない。

サブルーチン*BBT

表 1 中の式(2)と式(3)で示した base 関数の計算に必要な b と \bar{b} およびこれらの温度に関する偏導関数の値を求める。

サブルーチン*BASEDT

密度と温度の値を base 関数に代入して, 圧力を密度と気体定数と温度の積で割って求められる圧縮係数 (変数名 ZBASE), 圧縮係数の $y (= bd_w/4)$ に関する偏導関数の値 (変数名 DZB), 圧力の温度に関する偏導関数の値を気体定数と絶対温度と密度の積で割った値 (変数名 DPDTB) を求める。さらに, ヘルムホルツエネルギー (変数名 AB), 内部エネルギー (変数名 UB), 定容熱容量 (変数名 CVB), エントロピー (変数名 SB) を求める。AB, UB, CVB, SB の計算値は気体定数あるいは気体定数と絶対温度の積で割って求められる無次元化された値である。

サブルーチン*QQTD

密度と温度を residual 関数に代入して, 圧力 (変数名 Q), 圧力の密度に関する偏導関数の値 (変数名 Q5), 圧力の温度に関する偏導関数の値 (変数名 DPDTR) を求める。さらに, ヘルムホルツエネルギー (変数名 AR), 内部エネルギー (変数名 UR), 定容熱容量 (変数名 CVR), エントロピー (変数名 SR) を求める。AR, UR, CVR, SR の計算値は気体定数あるいは気体定数と絶対温度の積で割って求められる無次元化された値である。純水の密度 d_w と係数 ρ_i (i は 37 から 40) から求められる $d_w/\rho_i - 1$ の絶対値が 10^{-10} 未満の時に表 2 中の line 19950 で, この値を 10^{-10} に取っている。これは Haar et al. (1984) が示した FORTRAN のプログラムでも同じである。

サブルーチン*THERMDT

サブルーチン*IDEALT, *BASEDT, *QQTD を用いてヘルムホルツエネルギー (変数名 AD), ギブスエネルギー (変数名 GD), エントロピー (変数名 SD), 内部エネルギー (変数名 UD), エンタルピー (変数名 HD), 定容熱容量 (変数名 CVDX), 定圧熱容量 (変数名 CPD), 圧縮係数 (変数名 Z), 圧力の密度に関する偏導関数の値 (変数名 DPDD), 圧力の温度に関する偏導関数の値 (変数名 DPDT) を計算する。Z と DPDD と DPDT 以外の値は気体定数あるいは気体定数と絶対温度の積で割って無次元化した値として求められている。なお, 定容熱容量に関する変数名を CVD にした方が他の変数名と調和的になるが, CVD は BASIC/98 の予約語になっているので使用していない。

表6 サブルーチン*PCORR, *PST, *CORR, *UNIT, *UNIT_CONVERT, *TTTT,
*WATER_PROPERTY, *BLOCKDATAでの計算内容

サブルーチン*PCORR

飽和蒸気圧と気液二相の密度を計算する。サブルーチン*PSTで飽和蒸気圧の近似値(変数名PS)を求めた後で、サブルーチン*CORRで液相と気相の密度推定値(変数名DLとDV)とギブスエネルギーを無次元化した値(変数名GLとGV)、GL-GVの値を計算し、DL、DV、DELGの計算結果から飽和蒸気圧の近似値に補正值 δp (変数名DP)を加える。飽和蒸気圧の近似値を p 、 p から求めることができる液相と気相の密度の値を $d_w(l)$ と $d_w(v)$ 、液相1g当たりのギブスエネルギーを G^{liquid} 、気相1g当たりのギブスエネルギーを G^{vapor} と表すと $p + \delta p$ の時の液相と気相のギブスエネルギーは、それぞれ、 $G^{liquid} + \delta p/d_w(l)$ と $G^{vapor} + \delta p/d_w(v)$ として表すことができる。密度は圧力変化に応じて変化するが δp が小さい場合には密度の変化量も小さいので、ギブスエネルギーの変化量を $\delta p/d_w(l)$ や $\delta p/d_w(v)$ と近似している。すると、気液二相が平衡状態にある時には $G^{liquid} + \delta p/d_w(l)$ は $G^{vapor} + \delta p/d_w(v)$ の値とほぼ等しい。したがって、 δp の値を $(G^{liquid} - G^{vapor})/(1/d_w(v) - 1/d_w(l))$ より求めることができる。 $p + \delta p$ を新たな飽和蒸気圧の近似値 p にして液相と気相の密度を計算し、密度の値から液相と気相の1g当たりのギブスエネルギーを求める。そして、先に記した方法で新たな補正值 δp を求める。この逐次近似計算を繰り返して $(G^{liquid} - G^{vapor})/RT$ の値(変数名DELG)が収束条件を満足した時に計算を終了する。計算プログラムではDELGの絶対値が0.00001より小さくなっていった時に補正值を加えた圧力条件で気液二相が平衡状態にあるとしている。この時の圧力を飽和蒸気圧(変数名PPP)、液相と気相の密度(変数名DLとDV)を気液二相の密度とする。DELGの絶対値が0.00001以上の時には、気液二相の密度の推定値と補正後の圧力(変数名PPP)を用いて再びサブルーチン*CORRでDELGを計算する。気液二相平衡状態に関するDELGの条件をHaar et al. (1984)は絶対値が 10^{-4} より小さい時としたが、本計算プログラムで臨界点付近の計算にもう一桁厳しい収束条件を課した。

サブルーチン*PST

温度から飽和蒸気圧の近似値を計算する。

サブルーチン*CORR

温度と圧力から液相と気相と液相の密度(変数名DLとDV)およびこれらのギブスエネルギー(変数名GLとGV)を計算する。GL-GVの値を変数DELGの値にする。646.3Kより高温の時には澁江(2005b)中に示した式と方法によって飽和蒸気圧と気液二相の密度を計算し、変数DELGの値を0にする。

サブルーチン*UNIT

温度、密度、圧力、エネルギーの入力単位と出力単位を指定する。計算過程で使用している単位は、圧力がMPa、密度が $g\ cm^{-3}$ 、エネルギーが $J\ g^{-1}$ である。

サブルーチン*UNIT_CONVERT

サブルーチン*THERMDTで求めた計算結果を指定した単位で出力するための換算を行う。

サブルーチン*TTTT

入力した温度をHGK式中で用いる絶対温度に換算する。

サブルーチン*WATER_PROPERTY

計算結果を印字する。

サブルーチン*BLOCKDATA

表4中で示したHGK式中で用いられている定数値や係数値を読み込む。あわせて単位を印字するための文字列NNT\$(I)、NND\$(I)、NNP\$(I)、NNH\$(I)、文字変数A1\$, A2\$, A3\$, A4\$を読み込む。

4 計算の手順

温度と密度を入力する場合には次の手順で計算する。温度をサブルーチン *TTTT で絶対温度に換算してサブルーチン *BBT で base 関数の計算に必要なパラメータを求める。その後、サブルーチン *THERMDT に入って base 関数, residual 関数, ideal gas 関数から計算できる熱力学的性質の計算値を求める。サブルーチン *THERMDT とこのサブルーチン内で呼び出している3つのサブルーチン *IDEALT, *BASEDT, *QQTD でヘルムホルツエネルギー, ギブスエネルギー, 内部エネルギー, エントロピー, エンタルピー, 定圧熱容量, 定容熱容量の値を求めている。この時の値は, 気体定数と絶対温度の積あるいは気体定数で割って無次元化した値である。サブルーチン *THERMDT での計算が終了すると, サブルーチン *UNIT_CONVERT で熱力学的性質の単位を出力単位に変換する。この時, サブルーチン *THERMDT で求められた値に気体定数と絶対温度の積あるいは気体定数をかけあわせている。あわせて圧力の単位を出力単位に変換する。最後にサブルーチン *WATER_PROPERTY で結果を印字する。

温度と圧力を入力して計算する場合には逐次近似計算を行って求められた密度から熱力学的性質を求める。まず入力温度をサブルーチン *TTTT で絶対温度に変換して臨界温度 (HGK 式では647.126K) と比較する。臨界温度よりも高温の場合には, 温度と圧力から密度の値が一通りに決まる。この時の密度の初期推定値を $p/0.4T$ とおく。臨界温度よりも低温の場合には, 圧力が飽和蒸気圧と等しい場合と等しくない場合に分けておく。飽和蒸気圧と等しくない時にも密度の値が一通りに決まる。飽和蒸気圧に等しい時には, 密度の値が二通り求められる (液相の密度と気相の密度)。そこで, 臨界温度より低温の場合には飽和蒸気圧と気液二相の密度の計算をサブルーチン *PCORR で行う。

飽和蒸気圧 p_{sat} と入力圧力 p が, $-0.00005 \leq (p - p_{\text{sat}}) / p_{\text{sat}} \leq 0.00005$ の関係式を満たす場合には, 入力圧力が飽和蒸気圧であるとみなして液相と気相の性質を計算する。液相と気相の密度はサブルーチン *PCORR で求めている。温度と密度を求めることができたので先に記した手順で液相と気相の熱力学的性質を求める。

p の値が p_{sat} よりも大きい場合には飽和蒸気圧条件下での液相の密度を初期推定値にし, p_{sat} よりも小さい場合には $p/0.4T$ を気相の密度の初期推定値にする。密度の初期推定値をサブルーチン *DFIND で読み込んで, 密度の逐次近似計算を行う。密度の推定値から計算できる圧力が入力圧力と許容範囲内で等しくなった時に密度の計算を終了する。そして, 温度と密度の値から先に記した手順で熱力学的性質を求める。

5 計算値

三重点での液相の内部エネルギーとエントロピーの計算値は, それぞれ $-1.9782 \times 10^{-7} \text{ J g}^{-1}$ と $7.6930 \times 10^{-9} \text{ J g}^{-1} \text{ K}^{-1}$ であった。UREF と SREF の値を調節することで, いずれも値がほぼ0になった。

本計算プログラムによる計算値を Haar 達 (Haar et al., 1984) が蒸気表として p. 9から p. 237に示した結果 (温度—圧力—密度—エンタルピー—内部エネルギー—エントロピー—定圧熱容量の関係) と比較すると, 計算結果はほぼ完全に蒸気表を再現した。

Haar 達は温度—圧力条件に応じて HGK 式の不確かさを違えている。臨界点付近で最も計算値の有効桁数が少なくなる (密度の計算値は3桁) としている。本計算プログラムによる計算結果は蒸気表中の数値と臨界点付近で4桁目において食い違うことがある。ただし, 350℃以下では10kb まで数表値中の値の末尾の桁がごくたまに1食い違うことがあるだけである。

6 まとめ

Haar et al. (1984) が与えた純水の状態方程式を用いる計算プログラムを新たに作成した。水の三重点で液相の内部エネルギーとエントロピーの値を0になるように Haar et al. (1984) が与えた定数値を改めた。そして, 作成した計算プログラム中で使用しているサブルーチンの内容や計算手順を記述した。

7 文献

- Haar, L., Gallagher, J. S., and Kell, G. S. (1984) NBS/NRC Steam Tables. Hemisphere Publishing, 320pp.
- Kestin, J., Sengers, J. V., Kamgar-Parsi, B., and Levelt Sengers, J. M. H. (1984) J. Phys. Chem. Ref. Data, **13**, 175-183.
- Mao, S. and Duan, Z. (2008) The P, V, T, x properties of binary aqueous chloride solutions up to $T = 573 \text{ K}$ and 100 MPa. J. Chem. Thermodyn., **40**, 1046-1063.
- Pitzer, K. S. (1995) Thermodynamics. McGraw-Hill, 626pp.
- 澁江靖弘 (2005a) 気液二相平衡条件下での水の熱力学的性質を計算するプログラム— Haar et al. (1984) の式を用いて—。兵庫教育大学研究紀要, **26**, 105-117.
- 計算プログラムの部分的な修正を澁江 (2005b, p. 145) 中の表2と澁江 (2007a, p. 122) 中の表3で示している。これら以外に計算プログラムを示している p. 115の上から19行目中の $8858.843000000001/T$ は $8858.843\#/T$ に修正をする必要がある。また, このページの最下行に誤りがある。正しくは次の通りである。
 $23650 \text{ TAU} = .657128\# * (1\# - T/647.126\#)^{.325\#}$
 また, 表1-1中で示した b を与える式には誤りがある。ただし, 澁江 (2005a) 中の計算プログラムにおけ

る該当箇所には誤りがない。そして、図1として示した計算のフローシートに誤りがある。フローシート中で「計算を続けるか」に関する判断で2つの出力線に付けた「はい」と「いいえ」のラベルを入れ替える必要がある。

澁江靖弘 (2005b) 水の熱力学的性質を計算するプログラム - Haar et al. (1984) の式を用いて - 兵庫教育大学研究紀要, **27**, 143-154.

計算プログラムの部分的な修正を澁江 (2007a, p. 122) 中の表3で示している。これ以外に p.153の上から17行目に誤りがある。正しくは次の通りである。

$$18800 \text{ TAUC} = .657128\# * (1\# - T/647.126\#)^{\wedge}.325\#$$

澁江靖弘 (2007a) 300℃, 1000bar, 6 mol/kg までの塩化ナトリウム水溶液の密度を計算するプログラム - Pitzer-Peiper-Busey 式を用いて - 兵庫教育大学研究紀要, **30**, 115-128.

澁江靖弘 (2007b) 300℃, 1000bar, 濃度6 mol/kg までの塩化ナトリウム水溶液の熱力学的性質を計算するプログラム - Pitzer-Peiper-Busey 式を用いて - 兵庫教育大学研究紀要, **31**, 83-92.

表2中の下から4行目中の過剰エンタルピーは相対モルエンタルピーの意味で用いられている。

澁江靖弘 (2012a) 300℃から410℃における塩化カリウム - 水系気液平衡. 兵庫教育大学研究紀要, **40**, 79-91.

式(11)から式(13)に誤りがある。正しくは次の通りである。

$$-10^{-5} \leq y^{\text{vapor}(i+1)} / y^{\text{vapor}(i)} - 1 \leq 10^{-5} \quad (11)$$

$$-10^{-5} \leq d^{\text{vapor}(i+1)} / d^{\text{vapor}(i)} - 1 \leq 10^{-5} \quad (12)$$

$$-10^{-5} \leq d^{\text{liquid}(i+1)} / d^{\text{liquid}(i)} - 1 \leq 10^{-5} \quad (13)$$

計算プログラムを85ページから91ページ中で示している。85ページの下から19行目中で使用している変数 WTL を変数 WLKCL に訂正する必要がある。88ページの上から19行目中で使用している変数 DIFMYUKCL を変数 DIFMYUS に訂正する必要がある。ただし、これらを修正しなくとも正しい計算結果になる。その他にも不要な変数や演算が計算プログラムに含まれているが、これらは誤りではないのでここでは記さない。

澁江靖弘 (2012b) 250℃から600℃における塩化ナトリウム - 水系気液平衡. 兵庫教育大学研究紀要, **41**, 57-68.

式(11)から式(13)に誤りがある。正しくは次の通りである。

$$-10^{-5} \leq y^{\text{vapor}(i+1)} / y^{\text{vapor}(i)} - 1 \leq 10^{-5} \quad (11)$$

$$-10^{-5} \leq d^{\text{vapor}(i+1)} / d^{\text{vapor}(i)} - 1 \leq 10^{-5} \quad (12)$$

$$-10^{-5} \leq d^{\text{liquid}(i+1)} / d^{\text{liquid}(i)} - 1 \leq 10^{-5} \quad (13)$$

計算プログラムを61ページから63ページ中で示している。61ページの下から11行目中で使用している変数 WTL を変数 WLNACL に訂正する必要がある。ただ

し、これを修正しなくとも正しい計算結果になる。その他にも不要な変数や演算が計算プログラムに含まれているが、これらは誤りではないのでここでは記さない。

Wagner, W. and Pruß, A. (2002) The IAPWS formulation 1995 for the thermodynamic properties of ordinary water substance for general and scientific use. *J. Phys. Chem. Ref. Data*, **31**, 387-535.

Wagner, W., Cooper, J. R., Dittmann, A., Kojima, J., Kretzschmar, H.-J., Kruse, A., Mareš, R., Oguchi, K., Sato, H., Stöcker, I., Šifner, O., Takaishi, Y., Tanishita, I., Trübenbach, J., and Willkommen, Th. (2000) The IAPWS industrial formulation 1997 for the thermodynamic properties of water and steam. *Trans. ASME J. Eng. Gas Turbines Power*, **122**, 150-182.

Zein, D., Driesner, T., Scott, S., Sanchez-Valle, C., and Wagner, T. (2014) Volumetric properties of mixed electrolyte solutions at elevated temperatures and pressures. The systems CaCl₂-NaCl-H₂O and MgCl₂-NaCl-H₂O to 523.15 K, 70MPa, and ionic strength from (0.1 to 18) mol·kg⁻¹. *J. Chem. Eng. Data*, **59**, 2570-2588.